n-smooth   numbers are positive integers which have no prime factors  >  n.

The   n   (when using it in the expression)   n-smooth   is always prime,

there are   no   9-smooth numbers.

1   (unity)   is always included in n-smooth numbers.




2-smooth   numbers are non-negative powers of two.

5-smooth   numbers are also called   Hamming numbers.

7-smooth   numbers are also called    humble   numbers.



A way to express   11-smooth   numbers is:

  11-smooth  =  2i × 3j × 5k × 7m × 11p

           where     i, j, k, m, p ≥ 0  



Task

   calculate and show the first   25   n-smooth numbers   for   n=2   ───►   n=29
   calculate and show   three numbers starting with   3,000   n-smooth numbers   for   n=3   ───►   n=29
   calculate and show twenty numbers starting with  30,000   n-smooth numbers   for   n=503   ───►   n=521   (optional)


All ranges   (for   n)   are to be inclusive, and only prime numbers are to be used.

The (optional) n-smooth numbers for the third range are:   503,   509,   and   521.

Show all n-smooth numbers for any particular   n   in a horizontal list.

Show all output here on this page.



Related tasks

   Hamming numbers
   humble numbers


References

   Wikipedia entry:   Hamming numbers     (this link is re-directed to   Regular number).
   Wikipedia entry:   Smooth number
   OEIS entry:   A000079    2-smooth numbers or non-negative powers of two
   OEIS entry:   A003586    3-smooth numbers
   OEIS entry:   A051037    5-smooth numbers or Hamming numbers
   OEIS entry:   A002473    7-smooth numbers or humble numbers
   OEIS entry:   A051038   11-smooth numbers
   OEIS entry:   A080197   13-smooth numbers
   OEIS entry:   A080681   17-smooth numbers
   OEIS entry:   A080682   19-smooth numbers
   OEIS entry:   A080683   23-smooth numbers


import "/math" for Int
import "/big" for BigInt, BigInts
 
// cache all primes up to 521
var smallPrimes = Int.primeSieve(521)
var primes = smallPrimes.map { |p| BigInt.new(p) }.toList
 
var nSmooth = Fn.new { |n, size|
    if (n < 2 || n > 521) Fiber.abort("n must be between 2 and 521")
    if (size < 1) Fiber.abort("size must be at least 1")
    var bn = BigInt.new(n)
    var ok = false
    for (prime in primes) {
        if (bn == prime) {
            ok = true
            break
        }
    }
    if (!ok) Fiber.abort("n must be a prime number")
    var ns = List.filled(size, null)
    ns[0] = BigInt.one
    var next = []
    for (i in 0...primes.count) {
        if (primes[i] > bn) break
        next.add(primes[i])
    }
    var indices = List.filled(next.count, 0)
    for (m in 1...size) {
        ns[m] = BigInts.min(next)
        for (i in 0...indices.count) {
            if (ns[m] == next[i]) {
                indices[i] = indices[i] + 1
                next[i] = primes[i] * ns[indices[i]]
            }
        }
    }
    return ns
}
 
smallPrimes = smallPrimes.where { |p| p <= 29 }
for (i in smallPrimes) {
    System.print("The first 25 %(i)-smooth numbers are:")
    System.print(nSmooth.call(i, 25))
    System.print()
}
for (i in smallPrimes.skip(1)) {
    System.print("The 3,000th to 3,202nd %(i)-smooth numbers are:")
    System.print(nSmooth.call(i, 3002)[2999..-1])
    System.print()
}
for (i in [503, 509, 521]) {
    System.print("The 30,000th to 30,019th %(i)-smooth numbers are:")
    System.print(nSmooth.call(i, 30019)[29999..-1])
    System.print()
}